Sistemas de ecuaciones diferenciales con SAGE

Calcular los autovalores de una matriz:

```
A = matrix([[1,-2],[4,5]])
A.eigenvalues()
[3 - 2*I, 3 + 2*I]
```

Obtener los vectores propios asociados:

```
A.eigenvectors_right()
[(3 - 2*I, [(1, -1 + 1*I)], 1), (3 + 2*I, [(1, -1 - 1*I)], 1)]
```

Hallar la solución general del sistema de dos ecuaciones lineales $\begin{cases} x' = x - 2y \\ y' = 4x + 5y \end{cases}$

```
t=var('t')
x=function('x', t)
y=function ('y', t)
ec1=diff(x,t)==x-2*y
ec2=diff(y,t)==4*x+5*y
desolve_system([ec1, ec2], [x,y], ics=[0,1,1]);
[x(t) == -((x(0) + y(0))*sin(2*t) - cos(2*t)*x(0))*e^(3*t), y(t) ==
((2*x(0) + y(0))*sin(2*t) + cos(2*t)*y(0))*e^(3*t)]
```

Sage devuelve la solución general en función de los valores iniciales $x(0)$ e $y(0)$ (en lugar de las constantes arbitrarias C_1 y C_2). Obtener la solución del sistema que en el instante $t = 0$ pasa por $(1, 1)$:

```
sol=desolve_system([ec1,ec2],[x,y], ics=[0,1,1])
sol
[x(t) == -(2*sin(2*t) - cos(2*t))*e^(3*t), y(t) == (3*sin(2*t) +
cos(2*t))*e^(3*t)]
```

Representar la órbita:

```
sol1=sol[0].rhs()
sol2=sol[1].rhs()
f=parametric_plot((sol1, sol2), (t,-5,10))
f.show(xmin=-4, xmax=4, ymin=-4, ymax=4)
```



Obtener el límite de la órbita cuando t tiende a infinito

```
limit(sol1,t=oo), limit(sol2, t=oo);
(und, und)
```

lo que significa que el límite no existe (undefined). Para t tendiendo a menos infinito resulta:

```
limit(sol1,t=-oo), limit(sol2, t=-oo)
(0, 0)
```

Este resultado indica que la órbita sale del origen cuando el tiempo retrocede.

Representar algunas órbitas del sistema $\begin{cases} x' = -3x + 2y \\ y' = x - 2y \end{cases}$

```
t=var('t')
x=function('x', t)
y=function ('y', t)
ec1=diff(x,t)==-3*x+2*y
ec2=diff(y,t)==x-2*y
Lsol=[]
f2=plot([])
for i in srange(-4,4, 1):
    Lsol.append(desolve_system([ec1, ec2], [x,y], ics=[0,-1,i]))
    Lsol.append(desolve_system([ec1,ec2], [x,y], ics=[0,1,i]))
Lsol1=[]
Lsol2=[]
for i in srange(len(Lsol)):
    Lsol1.append(Lsol[i][0].rhs())
    Lsol2.append(Lsol[i][1].rhs())
    f2+=parametric_plot((Lsol1[i], Lsol2[i]), (t, -10, 10))
f2.show(xmin=-10, xmax=10, ymin=-10, ymax=10)
```



Ejercicio 1. Para el sistema $(a)\begin{cases} x' = x + y \\ y' = x - y \end{cases}$

1.1 Calcular los autovalores y vectores propios de la matriz asociada al sistema.

1.2 Obtener la solución general del sistema.

1.3 Hallar la solución particular que satisface $x(0) = 1$ e $y(0) = 1$ (en Sage ics=[0,1,1]) y representar la órbita en el intervalo de tiempo $-10 \leq t \leq 10$. Mostrar la gráfica en el marco (xmin=-10, xmax=10, ymin=-10, ymax=10).

1.4 Representar en una misma gráfica las órbitas que satisfacen las condiciones iniciales "for i in srange(-4,4, 1)" "ics=[0,-1,i]" y "ics=[0,1,i]".

1.5 Concluir sobre la estabilidad de las soluciones.

1.6 Realizar un análisis similar para los siguientes sistemas, cambiando las condiciones iniciales, el intervalo del tiempo, el marco de la gráfica, etc.

$$\text{(b)}\ \begin{cases} x' = -5x - 4y \\ y' = -2x - 4y \end{cases} \qquad \text{(c)}\ \begin{cases} x' = -x - 5y \\ y' = x - y \end{cases} \qquad \text{(d)}\ \begin{cases} x' = -x - 5y \\ y' = x + y \end{cases}$$

Modelos de competencia e interacción entre especies Supongamos que tenemos dos especies de seres vivos, una de las cuales es depredadora y otra presa. Sean $x(t)$ e $y(t)$ el número de presas y depredadores, respectivamente, en el instante t (contadas en miles de individuos). El tiempo t lo mediremos en años.
Vamos a analizar el comportamiento de las especies en un intervalo del tiempo suponiendo que la interacción entre ambas especies se rige por el sistema:

$$\begin{cases} x' = 3x - y \\ y' = 3x + y \\ x(0) = 3, y(0) = 2 \end{cases}$$

El modelo asume las siguientes hipótesis:

- La tasa de cambio de individuos en cada especie, sin interacción, es proporcional al número de individuos de la especie. Las constantes de proporcionalidad para presas y depredadores son 3 y 1, respectivamente.
- La tasa de disminución en la especie presa es proporcional al número de depredadores, siendo la constante de proporcionalidad 1 que se refleja con signo negativo porque es disminución.
- La tasa de aumento en la especie depredadora es proporcional al número de presas, siendo la constante de proporcionalidad 3.
- Inicialemnte hay 3.000 presas y 2.000 depredadores.

Ejercicio 2 Hallar, con SAGE, la solución del sistema y representar la órbita en el intervalo $0 \leq t \leq 1$. ¿Cuál es el comportamiento del sistema, según se refleja en la órbita? (aumento/disminución de depredadores y presas). Representa la órbita en el intervalo $0 \leq t \leq 2$. ¿Qué observas? Razona si el modelo de interacción de ambas especies es adecuado para un intervalo de tiempo de dos años.

Matriz de Jordan y exponencial

Podemos trabajar con distintos cuerpos para los coeficientes: QQ el cuerpo de los números racionales, RDF números reales de 64 bits. Calcular los autovalores, los vectores propios y la forma de Jordan de una matriz:

```
A=matrix(QQ, 3,3,[2,-1,-1,2,-1,-2,-1,1,2])
print A
[ 2 -1 -1]
```

```
[ 2 -1 -2]
[-1  1  2]
show(A.eigenvectors_right())
[(1,[(1,0,1),(0,1,2)],3)]
Ja=A.jordan_form()
[1 1|0]
[0 1|0]
[0 0|1]
```

Podemos obtener la matriz de paso:

```
P=A.jordan_form(transformation=True)[1]
print P
[ 1  1  1]
[ 2  0  0]
[-1  0  1]
```

Usando la matriz de paso y la matriz de Jordan podemos escribir una matriz fundamental del sistema diferencial $X'(T) = AX(t)$:

```
t=var('t')
exp(Ja*t)
[  e^t t*e^t     0]
[    0   e^t     0]
[    0     0   e^t]

 M=P*exp(Ja*t)
print M
[        e^t t*e^t + e^t         e^t]
[      2*e^t     2*t*e^t           0]
[       -e^t      -t*e^t         e^t]
```

Resolver el sistema:

```
var('t')
x=function('x',t)
y=function('y',t)
z=function('z',t)
ec1=diff(x,t)==2*x-y-z
ec2=diff(y,t)==2*x-y-2*z
ec3=diff(z,t)==-x+y+2*z
desolve_system([ec1,ec2,ec3], [x,y,z])
```

Calcular subespacios propios generalizados:

```
Id=identity_matrix(3, sparse=True);
kernel(A-Id)
Vector space of degree 3 and dimension 2 over Rational Field
Basis matrix:
[1 0 1]
[0 1 2]
```

Ejercicio 3 Para los siguientes sistemas, calcular con Sage una matriz fundamental usando la forma de Jordan y el cálculo de la exponencial matricial. Obtener los subespacios propios asociados a cada autovalor y observar la relación con los vectores de la matriz de paso.

$$\begin{cases} x' = 4x - y \\ y' = 3x + y - z \\ z' = x + z \end{cases} \qquad \begin{cases} x' = 2x + y - z \\ y' = x + 2y - z \\ z' = 8x - 2y - 2z \end{cases} \qquad \begin{cases} x' = u \\ y' = z \\ z' = -x - y \\ u' = 3x + 4y \end{cases}$$

Sistemas no lineales

Podemos resolver usando el método numérico de Euler el sistema no lineal $\begin{cases} x' = -y + x(1 - x^2 - y^2) \\ y' = x + y(1 - x^2 - y^2) \\ x(0) = 2, y(0) = 0 \end{cases}$

en el intervalo de tiempo $0 \leq t \leq 10$, con un paso de malla 0.1. En este caso Sage devuelve una lista de puntos $[t_k, x_k, y_k]$ y podemos representar las soluciones aproximadas $[t_k, x_k]$ y $[t_k, y_k]$ en una figura:

```
from sage.calculus.desolvers import eulers_method_2x2
t, x, y = PolynomialRing(QQ,3,"txy").gens()
f =  -y+x*(1-x^2-y^2); g = x+y*(1-x^2-y^2)
P=eulers_method_2x2(f,g, 0, 2, 0, 0.1, 10,algorithm="none")
Q1=[ [i,j] for i,j,k in P]
Q2=[[i,k] for i,j,k in P]
LP1=list_plot(Q1, color="red")
LP2=list_plot(Q2, color="green")
(LP1+LP2).show()
```



Dibujar la órbita del sistema:

```
Q=[[j,k] for i,j,k in P]
LP=list_plot(Q)
LP
```



Modelo predador-presa de Lotka-Volterra

Sean $x(t)$ e $y(t)$ el número de presas y depredadores, respectivamente, en el instante t (contadas en miles de individuos). El tiempo t lo mediremos en semanas. Vamos a analizar el comportamiento del siguiente modelo no lineal:

$$\begin{cases} x' = x - xy \\ y' = -\frac{1}{3}y + \frac{1}{3}xy \\ x(0) = 2, y(0) = 2 \end{cases}$$

- En ausencia de depredadores la población de presas crece a una tasa proporcional al número de individuos. Aquí la constante de proporcionalidad es 1.
- En ausencia de presas la población de depredadores disminuye a una tasa proporcional al número de individuos, siendo la constante de proporcionalidad $\frac{1}{3}$ (con signo negativo).
- Cuando las especies interactúan, la población de presas disminuye y la de los depredadores aumenta en relación con la frecuencia de los encuentros entre individuos de las dos especies, que supondremos es proporcional al producto xy. Suponemos que las constante de proporcionalidad para la disminución de presas es 1 y para el aumento de depredadores es $\frac{1}{3}$.
- Inicalmente hay 2.000 presas y 2.000 depredadores.

Ejercicio 4 Resolver, usando el método numérico de Euler, el sistema del Lotka-Volterra y representa en una figura las gráficas de los soluciones aproximadas $[t_k, x_k]$ y $[t_k, y_k]$ en el intervalo $0 \le t \le 50$, con un paso de malla 0.1. Comentar los intervalos de aumento/disminución de presas y depredadores. Dibuja la órbita del sistema y observa el comportamiento periódico.

Ejercicio 5 Resolver, usando el método numérico de Runge-Kutta (desolve-system-rk4), el sistema $\begin{cases} x' = -y - x(\sqrt{x^2 + y^2}) \\ y' = x - y\sqrt{x^2 + y^2} \\ x(0) = 3, y(0) = 3 \end{cases}$ en el intervalo de tiempo $0 \le t \le 10$, con un paso de malla 0,1. Representar las soluciones aproximadas $[t_k, x_k]$ y $[t_k, y_k]$ en una figura y dibujar la órbita del sistema.